*==========================================================================================================================================================
*
*	Local Projections
*
*==========================================================================================================================================================


set more off

qui {


*-------------------------------------------------------------------------------------------------------------
*	SETTINGS
*-------------------------------------------------------------------------------------------------------------

* Variables
local vars "exp_iron nor_ewo_sac bay_imp_iep"   

* irf horizon
local H 5
local F 0

* rhs controls lags
local L 2 

*-------------------------------------------------------------------------------------------------------------


*_________ GET DATA ____________
use "data/working files/prepared.dta", clear 


*____ DECLARE TIME SERIES _____
gen id =1
tsset id year, yearly


*_________________ LOCAL PROJECTIONS _________________

local count = 0
foreach v of local vars {
local count = `count'+1
n di "Variable `v'"

***_________ CONTROLS

local indep	" gdp_r_adj_gr cpi_sil_gr wage_sil_gr itsN_gr" 

forvalues i=0/`L'{
	local indep_`i' = subinstr("`indep'"," "," l`i'_",.)
	}
	
local rhs ""
forvalues i = 1/`L'{
	local rhs "`rhs'`indep_`i''"
	}
	
	local not l0_`v'_gr l0_itsN_gr l0_itsN
	local rhs: list rhs- not
	
	


	***_________ IMPULSE RESPONSE and CONFIDENCE INTERVAL variables
	gen irf_`v' = . if _n<=`H'+1
	
	gen se_`v' = . if _n<=`H'+1

	gen sig_`v' = . if _n<=`H'+1
	gen ci_up_`v' = . if _n<=`H'+1
	gen ci_lo_`v' = . if _n<=`H'+1
	gen ci_up90_`v' = . if _n<=`H'+1
	gen ci_lo90_`v' = . if _n<=`H'+1
			

	
	***_________ ESTIMATE COEFFICIENTS
	
	forvalues y = 0/`H' {
		n display "projection `y'"
		
		sum `v' 
		if (`r(N)' < 100) {
			local F = 0
			local LS = 1
			newey `v'_gr`y' l(1/`LS').`v'_gr l(-`F'/`LS').percloss_netN, lag(5) force
		}

		sum `v'
		if (`r(N)' >= 100) {
			local F = 5
			local LS = 2
			newey `v'_gr`y' l(1/`LS').`v'_gr l(-`F'/`LS').perclossN_vellon l(-`F'/`LS').perclossN l(-`F'/`LS').percloss_salvN `rhs' l(-`F'/`LS').percloss_pirN l(0/`LS').temperature l(0/`LS').warbeu l(0/`LS').warbciv, lag(5) force
		}
		
		* store IMPULSE RESPONSES and STANDARD ERRORS	
		sum `v'
		if (`r(N)' >= 100) {
			lincom perclossN 
			}
		else {
			lincom percloss_netN
			}
		replace irf_`v' = r(estimate) if _n==`y'+1 /* coefficients */
		replace se_`v' = r(se) if _n==`y'+1 /* standard errors */
		

		} //horizon			
	} // vars

		

	
local count = 0
foreach v of local vars {
local count = `count'+1
n di "Variable `v'"

	***_________ CONFIDENCE INTERVALS
	replace ci_up_`v' = irf_`v' + 1*se_`v' if _n <= `H'+1
	replace ci_lo_`v' = irf_`v' - 1*se_`v' if _n <= `H'+1
	replace ci_up90_`v' = irf_`v' + 1.64*se_`v' if _n <= `H'+1
	replace ci_lo90_`v' = irf_`v' - 1.64*se_`v' if _n <= `H'+1
	
	replace sig_`v' =  irf_`v' if ci_up90_`v' < 0 | ci_lo90_`v' > 0

}



* plot prerequisites
gen time = _n-1 if(_n<=`H'+1) 
drop if time > `H'
keep time irf_* ci_up_* ci_lo_* ci_up90_* ci_lo90_* se_* sig_*
order irf_* ci_up_* ci_lo_* time ci_up90_* ci_lo90_* sig_*
drop if time == . 
gen zero = 0 if time!=.


local lsize "5.5"
local tsize "6"
local legsize "5.5"

local zline "medthin"
local irfline "medlarge"
local msize "vlarge"

		
		//Real exports
twoway  (line zero time, lcolor(black) lwidth(`zline')) ///
		(scatter irf_exp_iron time, color(black) msymbol(circle_hollow) msize(`msize')) ///
		(line irf_exp_iron time, lcolor(black) lpattern(dot) lwidth(`irfline')) ///
		(scatter irf_nor_ewo_sac time, color(gs8) msymbol(diamond_hollow) msize(`msize')) ///
		(line irf_nor_ewo_sac time, lcolor(gs8) lpattern(dot) lwidth(`irfline')) ///
		(scatter sig_exp_iron time, color(black) msymbol(circle) msize(`msize')) ///
		(scatter sig_nor_ewo_sac time, color(gs8) msymbol(diamond) msize(`msize')), ///
		title("Real exports", size(`tsize')) xtitle("Year", size(`lsize')) ytitle("Percent", height(2) size(`lsize')) xlabel(,labsize(`lsize')) ylabel(, format(%9.1f) nogrid angle(0) labsize(`lsize')) ///
		legend(size(`legsize') symxsize(*.5) order(2 4) col(1) pos(7) ring(0) lab(2 "Iron") lab(4 "Wool") region(color(white%0))) ///
		graphregion(color(white)) plotregion(color(white)) name(g999, replace) nodraw
		
	

//Imports from Bayonne 
twoway  (line zero time, lcolor(black) lwidth(`zline')) ///
		(scatter irf_bay_imp_iep time, color(black) msymbol(circle_hollow) msize(`msize')) ///
		(line irf_bay_imp_iep time, lcolor(black) lpattern(dot) lwidth(`irfline')) ///
		(scatter sig_bay_imp_iep time, color(black) msymbol(circle) msize(`msize')), ///
		title("Nominal imports from Bayonne", size(`tsize')) xtitle("Year", size(`lsize')) ytitle("Percent", height(2) size(`lsize')) xlabel(,labsize(`lsize')) ylabel(, format(%9.1f) nogrid angle(0) labsize(`lsize')) ///
		legend(off) ///
		graphregion(color(white)) plotregion(color(white)) name(g8008, replace) nodraw

		
graph combine g999 g8008, name(LP_disag_trade, replace) cols(2) imargin(medsmall) plotregion(color(white)) graphregion(color(white)) 
graph display LP_disag_trade, ysize(4.5) xsize(10)  
graph export "results/FigureB3.pdf", replace 	

} // qui


*_________________________________________

			

			
